

## 钢管下料问题

某钢管零售商从钢管厂进货，将钢管按照顾客的要求切割后售出，从钢管厂进货时得到的原料钢管都是19m。

(1) 现在一客户需要50根4m、20根6m和15根8m的钢管。应如何下料最节省？

(2) 零售商如果采用的不同切割模式太多，将会导致生产过程的复杂化，从而增加生产和管理成本，所以该零售商规定采用的不同切割模式不能超过3种。此外，该客户除需要(1)中的三种钢管外，还需要10根5m的钢管。应如何下料最节省。

### 问题（1）分析与模型建立

19m 的钢管切割为 4m、6m、8m 的钢管的模式采用向量  $(k_1, k_2, k_3)$  表示。

所有模式相当于求解不等式方程的整数解：

$$4k_1 + 6k_2 + 8k_3 \leq 19$$

要求剩余材料  $r = 19 - (4k_1 + 6k_2 + 8k_3) < 4$ 。

表 1 钢管切割模式

模式	4m	6m	8m	余料(m)
1	4	0	0	3
2	3	1	0	1
3	2	0	1	3
4	0	0	2	3
5	0	3	0	1
6	1	1	1	1
7	1	2	0	3

**决策变量** 用  $x_i$  表示按照第  $i$  种模式( $i=1,2,\cdots, 7$ )切割的原料钢管的根数。

**决策目标** 以切割原料钢管的总根数最少为目标, 则有

$$\min z_1 = \sum_{i=1}^7 x_i$$

以切割后剩余的总余料最小为目标, 设第  $i$  种模式的余料为  $r_i$  米。则由表 1 可得

$$\min z_2 = \sum_{i=1}^7 r_i x_i$$

设第  $i$  种切割模式下 4 米长的钢管  $a_i$  根,

6 米长的钢管  $b_i$  根, 8 米长的钢管  $c_i$  根,

则约束条件有:

4 米长的钢管至少 50 根, 有 
$$\sum_{i=1}^7 a_i x_i \geq 50$$

6 米长的钢管至少 20 根, 有 
$$\sum_{i=1}^7 b_i x_i \geq 20$$

8 米长的钢管至少 15 根, 有 
$$\sum_{i=1}^7 c_i x_i \geq 15$$

总模型为:

$$\min z_1 = \sum_{i=1}^7 x_i$$

$$\min z_2 = \sum_{i=1}^7 r_i x_i$$

$$s.t. \begin{cases} \sum_{i=1}^7 a_i x_i \geq 50 \\ \sum_{i=1}^7 b_i x_i \geq 20 \\ \sum_{i=1}^7 c_i x_i \geq 15 \\ x_i \text{取整}, i = 1, 2, \dots, 7 \end{cases}$$

解得：

$$x_1 = 0, x_2 = 15, x_3 = 0, x_4 = 5, x_5 = 0, x_6 = 5, x_7 = 0$$

目标值  $z_1=25$ ,  $z_2=35$ 。

即 15 根钢管采用切割模式 2：3 根 4m，1 根 6m，余料 1m。

5 根钢管采用切割模式 4：2 根 8m，余料 3m。

5 根钢管采用切割模式 6：1 根 4m，1 根 6m，1 根 8m，余料 1m。

切割模式采用了 3 种，使用钢管 25 根，余料为 35m。

固定  $z_1=25$ ，求  $z_2$  最小结果一样。

说明在该问题中当使用钢管数最少时，余料也最少。

**LINGO程序：**

model:

sets:

model/1..7/:a,b,c,r,x;

endsets

```
data:
a=4,3,2,0,0,1,1;
b=0,1,0,0,3,1,2;
c=0,0,1,2,0,1,0;
r=3,1,3,3,1,1,3;
enddata
min=z1;
z1=@sum(model(i):x(i));!钢管总数;
z2=@sum(model(i):r(i)*x(i));!余料;
@sum(model(i):a(i)*x(i))>=50;!4米长钢管约束;
@sum(model(i):b(i)*x(i))>=20;!6米长钢管约束;
@sum(model(i):c(i)*x(i))>=15;!8米长钢管约束;
@for(model(i):@gin(x(i)));
end
```

## 问题（2）模型建立

19m 的钢管切割为 4m、6m、8m、5m 的钢管的模式，  
所有模式相当于求解不等式方程的整数解。

$$4k_1 + 6k_2 + 8k_3 + 5k_4 \leq 19$$

要求剩余材料  $r = 19 - (4k_1 + 6k_2 + 8k_3 + 5k_4) < 4$ 。

求出所有模式的Matlab程序见下：



```
number=0;
for k1=0:4
    for k2=0:3
        for k3=0:2
            for k4=0:3
                r=19-(4*k1+6*k2+8*k3+5*k4);
                if(r>=0)&&(r<4)
                    number=number+1;
                    fprintf('%2d %2d %2d %2d %2d %2d\n',number,k1,k2,k3,k4,r);
                end
            end
        end
    end
end
end
```

表 2 钢管切割模式

模式	4m	6m	8m	5m	余料(m)
1	0	0	1	2	1
2	0	0	2	0	3
3	0	1	0	2	3
4	0	1	1	1	0
5	0	2	0	1	2
6	0	3	0	0	1
7	1	0	0	3	0
8	1	0	1	1	2
10	1	2	0	0	3
11	2	0	0	2	1
12	2	0	1	0	3
13	2	1	0	1	0
14	3	0	0	1	2
15	3	1	0	0	1
16	4	0	0	0	3

**决策变量** 用  $x_i$  表示按照第  $i$  种模式( $i=1,2,\cdots, 16$ )切割的原料钢管的根数。

**决策目标** 以切割原料钢管的总根数最少为目标, 则有

$$\min z_1 = \sum_{i=1}^{16} x_i$$

以切割后剩余的总余料量最小为目标, 设第  $i$  种模式的余料为  $r_i$  米。则得

$$\min z_2 = \sum_{i=1}^{16} r_i x_i$$

设第  $i$  种切割模式下 4 米长的钢管  $a_i$  根, 6 米长的钢管  $b_i$  根,

8 米长的钢管  $c_i$  根, 5 米长的钢管  $d_i$  根。则约束条件有:

4 米长的钢管至少 50 根，有

$$\sum_{i=1}^{16} a_i x_i \geq 50$$

6 米长的钢管至少 20 根，有

$$\sum_{i=1}^{16} b_i x_i \geq 20$$

8 米长的钢管至少 15 根，有

$$\sum_{i=1}^{16} c_i x_i \geq 15$$

5 米长的钢管至少 10 根，有

$$\sum_{i=1}^{16} d_i x_i \geq 10$$

最多使用 3 种切割模式，增设 0-1 变量  $y_i, i = 1, 2, \dots, 16$ 。

$$\text{有 } \sum_{i=1}^{16} y_i \leq 3$$

当  $y_i = 0$  时,  $x_i = 0$ , 表示不使用第  $i$  种切割模式;

则  $x_i \leq M \cdot y_i$

当  $y_i = 1$  时,  $x_i \geq 1$ , 表示使用第  $i$  种切割模式。

因此有:  $x_i \geq y_i, x_i \leq M \cdot y_i, i = 1, 2, \dots, 16$

其中  $M$  足够大, 如这里取 1000。

$$\min z_1 = \sum_{i=1}^{16} x_i$$

$$\min z_2 = \sum_{i=1}^{16} r_i \cdot x_i$$

总模型：

$$s.t. \begin{cases} \sum_{i=1}^{16} a_i x_i \geq 50 & \sum_{i=1}^{16} b_i x_i \geq 20 \\ \sum_{i=1}^{16} c_i x_i \geq 15 & \sum_{i=1}^{16} d_i x_i \geq 10 \\ x_i \leq M \cdot y_i, i = 1, 2, \dots, 16 \\ x_i \geq y_i, i = 1, 2, \dots, 16 \\ \sum_{i=1}^{16} y_i \leq 3 \\ x_i \text{取整}, i = 1, 2, \dots, 16 \\ y_i = 0 \text{或} 1, i = 1, 2, \dots, 16 \end{cases}$$

解得：

1) 当使用钢管数  $z_1$  最小时，求得的解为：

$$x_2 = 8, x_{13} = 10, x_{15} = 10, \text{ 其余为 } 0。$$

目标值  $z_1=28, z_2=34。$

即8根钢管采用切割模式2:2根8m,余料3m.

10根钢管采用切割模式13:2根4m,1根6m,1根5m,余料为0

10根钢管采用切割模式15:3根4m,1根6m,余料1m.

切割模式采用了3种,使用钢管 $z_2=28$ 根,余料为 $z_1=34m。$

固定 $z_1=28$ ,求 $z_2$ 最小结果一样。

说明在该问题中当使用钢管数最少时,余料也最少.

LINGO程序为:

model:

sets:

model/1..16/:a,b,c,d,r,x,y;

endsets

data:

a=0,0,0,0,0,0,1,1,1,1,2,2,2,3,3,4;

b=0,0,1,1,2,3,0,0,1,2,0,0,1,0,1,0;

c=1,2,0,1,0,0,0,1,1,0,0,1,0,0,0,0;

d=2,0,2,1,1,0,3,1,0,0,2,0,1,1,0,0;

r=1,3,3,0,2,1,0,2,1,3,1,3,0,2,1,3;

enddata



```
min=z1;  
z1=@sum(model(i):x(i));!钢管总数;  
z2=@sum(model(i):r(i)*x(i));!余料;  
@sum(model(i):a(i)*x(i))>=50;!4米长钢管约束;  
@sum(model(i):b(i)*x(i))>=20;!6米长钢管约束;  
@sum(model(i):c(i)*x(i))>=15;!8米长钢管约束;  
@sum(model(i):d(i)*x(i))>=10;!5米长钢管约束;  
@for(model(i):x(i)>=y(i));  
@for(model(i):x(i)<=1000*y(i));  
@sum(model(i):y(i))<=3;  
@for(model(i):@gin(x(i)));  
@for(model(i):@bin(y(i)));  
end
```

谢 谢！