

## 第 2 讲 数学规划与目标规划

司守奎

烟台市, 海军航空大学

Email: sishoukui@163.com

### 2.1 数学规划应用问题举例

#### 2.1.1 线性规划

例 2.1 某部门在今后五年内考虑给下列项目投资, 已知:

项目 A, 从第一年到第四年每年年初需要投资, 并于次年末回收本利 115%;

项目 B, 从第三年初需要投资, 到第五年末能回收本利 125%, 但规定最大投资额不超过 4 万元;

项目 C, 第二年初需要投资, 到第五年末能回收本利 140%, 但规定最大投资额不超过 3 万元;

项目 D, 五年内每年年初可购买公债, 于当年末归还, 并加利息 6%。

该部门现有资金 10 万元, 问它应如何确定给这些项目每年的投资额, 使到第五年末拥有的资金的本利总额为最大?

解 用  $j=1,2,3,4$  分别表示项目 A,B,C,D, 用  $x_{ij}$  ( $i=1,2,3,4,5$ ) 分别表示第  $i$  年年初给项目 A,B,C,D 的投资额。根据给定的条件, 对于项目 A 存在变量:  $x_{11}, x_{21}, x_{31}, x_{41}$ ; 对于项目 B 存在变量:  $x_{32}$ ; 对于项目 C 存在的变量:  $x_{23}$ ; 对于项目 D 存在变量:  $x_{14}, x_{24}, x_{34}, x_{44}, x_{54}$ 。

该部门每年应把资金全部投出去, 手中不应当有剩余的呆滞资金。

第一年:  $x_{11} + x_{14} = 100000$ 。

第二年初部门拥有的资金是项目 D 在第一年末回收的本利, 于是第二年的投资分配为  $x_{21} + x_{23} + x_{24} = 1.06x_{14}$ 。

第三年初部门拥有的资金是项目 A 第一年投资及项目 D 第二年投资中回收的本利总和。于是第三年的资金分配为

$x_{31} + x_{32} + x_{34} = 1.15x_{11} + 1.06x_{24}$ 。

类似地可得

第四年:  $x_{41} + x_{44} = 1.15x_{21} + 1.06x_{34}$ 。

第五年:  $x_{54} = 1.15x_{31} + 1.06x_{44}$ 。

此外, 项目 B,C 的投资额限制, 即

$x_{32} \leq 40000$ ,  $x_{23} \leq 30000$ 。

问题是要求在第五年末该部门手中拥有的资金额达到最大, 目标函数可表示为

$\max z = 1.15x_{41} + 1.40x_{23} + 1.25x_{32} + 1.06x_{54}$ 。

综上所述, 建立如下的线性规划模型:

$$\begin{aligned} \max z &= 1.15x_{41} + 1.40x_{23} + 1.25x_{32} + 1.06x_{54}, \\ \text{s.t. } &x_{11} + x_{14} = 100000, \\ &x_{21} + x_{23} + x_{24} = 1.06x_{14}, \\ &x_{31} + x_{32} + x_{34} = 1.15x_{11} + 1.06x_{24}, \\ &x_{41} + x_{44} = 1.15x_{21} + 1.06x_{34}, \\ &x_{54} = 1.15x_{31} + 1.06x_{44}, \\ &x_{32} \leq 40000, \quad x_{23} \leq 30000, \\ &x_{ij} \geq 0, \quad i=1,2,3,4,5; j=1,2,3,4. \end{aligned}$$

计算的 LINGO 程序如下。

model:

sets:

row/1..5/;

col/1..4/;

```

link(row,col):x;
endsets
max=1.15*x(4,1)+1.4*x(2,3)+1.25*x(3,2)+1.06*x(5,4);
x(1,1)+x(1,4)=100000;
x(2,1)+x(2,3)+x(2,4)=1.06*x(1,4);
x(3,1)+x(3,2)+x(3,4)=1.15*x(1,1)+1.06*x(2,4);
x(4,1)+x(4,4)=1.15*x(2,1)+1.06*x(3,4);
x(5,4)=1.15*x(3,1)+1.06*x(4,4);
x(3,2)<40000; x(2,3)<30000;
end

```

例 2.2 捷运公司在下一年度的 1~4 月的 4 个月内拟租用仓库堆放物资。已知各月份所需仓库面积列于表 2.1。仓库租借费用随合同期而定，期限越长，折扣越大，具体数字见表 2.1。租借仓库的合同每月初都可办理，每份合同具体规定租用面积和期限。因此该公司可根据需要，在任何一个月初办理租借合同。每次办理时可签一份合同，也可签若干份租用面积和租借期限不同的合同，试确定该公司签订租借合同的最优决策，目的是使所付租借费用最小。

表 2.1		100m <sup>2</sup>		
月份	1	2	3	4
所需仓库面积	15	10	20	12
合同租借期限	1 个月	2 个月	3 个月	4 个月
合同期内的租费	2800	4500	6000	7300

解 设变量  $x_{ij}$  表示捷运公司在第  $i$  ( $i=1, \dots, 4$ ) 个月初签订的租借期为  $j$  ( $j=1, \dots, 4$ ) 个月的仓库面积的合同 (单位为 100m<sup>2</sup>)。因 5 月份起该公司不需要租借仓库，故  $x_{24}, x_{33}, x_{34}, x_{42}, x_{43}, x_{44}$  均为零。该公司希望总的租借费用为最小，故有如下数学模型：

$$\begin{aligned}
 & \text{(目标函数)} \quad \min z = 2800(x_{11} + x_{21} + x_{31} + x_{41}) + 4500(x_{12} + x_{22} + x_{32}) \\
 & \quad \quad \quad + 6000(x_{13} + x_{23}) + 7300x_{14} \\
 & \text{s.t. (约束条件)} \quad \begin{cases} x_{11} + x_{12} + x_{13} + x_{14} \geq 15, \\ x_{12} + x_{13} + x_{14} + x_{21} + x_{22} + x_{23} \geq 10, \\ x_{13} + x_{14} + x_{22} + x_{23} + x_{31} + x_{32} \geq 20, \\ x_{14} + x_{23} + x_{32} + x_{41} \geq 12, \\ x_{ij} \geq 0, i=1, \dots, 4; j=1, \dots, 4. \end{cases}
 \end{aligned}$$

这个模型中的约束条件分别表示当月初签订的租借合同的面积加上该月前签订的未到期的合同的租借面积总和，应不少于该月所需的仓库面积。

求得的最优解为  $x_{11} = 3$ ,  $x_{31} = 8$ ,  $x_{14} = 12$ ，其它变量取值均为零，最优值  $z^* = 118400$ 。

计算的 Lingo 程序如下：

```

model:
sets:
num/1..4/;
link(num,num):x;
endsets
min=2800*(x(1,1)+x(2,1)+x(3,1)+x(4,1))+4500*(x(1,2)+x(2,2)+x(3,2))+6000*(x(1,3)+x(2,3))
+7300*x(1,4);
x(1,1)+x(1,2)+x(1,3)+x(1,4)>15;

```

```

x(1,2)+x(1,3)+x(1,4)+x(2,1)+x(2,2)+x(2,3)>10;
x(1,3)+x(1,4)+x(2,2)+x(2,3)+x(3,1)+x(3,2)>20;
x(1,4)+x(2,3)+x(3,2)+x(4,1)>12;
end

```

也可以编写如下数据单独列出的 LINGO 程序：

```

model:
sets:
num/1..4/:c,d;
link(num,num):x;
endsets
data:
d=15 10 20 12;
c=2800 4500 6000 7300;
enddata
min=@sum(link(i,j)|i#le#5-j:c(j)*x(i,j));
@for(num(k):@sum(link(i,j)|i#le#k #and# j#ge#k+1-i#and#j#le#5-i:x(i,j))>d(k));
end

```

### 2.1.2 整数规划

例 2.3 求解指派矩阵  $C$  的指派问题。

$$C = \begin{bmatrix} 12 & 7 & 9 & 7 & 9 \\ 8 & 10 & 6 & 6 & 6 \\ 7 & 17 & 12 & 14 & 9 \\ 15 & 14 & 6 & 6 & 10 \\ 4 & 10 & 7 & 7 & 9 \end{bmatrix}$$

这里  $C = (c_{ij})_{5 \times 5}$  中的  $c_{ij}$  表示第  $i$  个人干第  $j$  项工作花费的时间（单位：小时）。

解 首先建立指派问题的 0-1 整数规划模型，引进 0-1 变量

$$x_{ij} = \begin{cases} 1, & \text{第 } i \text{ 个人干第 } j \text{ 项工作,} \\ 0, & \text{第 } i \text{ 个人不干第 } j \text{ 项工作.} \end{cases}$$

建立的 0-1 整数规划模型如下

$$\begin{aligned} \min & \sum_{i=1}^5 \sum_{j=1}^5 c_{ij} x_{ij}, \\ \text{s.t.} & \begin{cases} \sum_{j=1}^5 x_{ij} = 1, i = 1, 2, \dots, 5, \\ \sum_{i=1}^5 x_{ij} = 1, j = 1, 2, \dots, 5, \\ x_{ij} = 0 \text{ 或 } 1, i, j = 1, 2, \dots, 5. \end{cases} \end{aligned}$$

利用 LINGO 程序求得指派问题的解为

$$\begin{bmatrix} 1 & 2 & 3 & 4 & 5 \\ 2 & 4 & 5 & 3 & 1 \end{bmatrix},$$

即第 1 个人干第 2 项工作，第 2 个人干第 4 项工作，第 3 个人干第 5 项工作，第 4 个人干第 3 项工作，第 5 个人干第 1 项工作。

计算的 LINGO 程序如下：

```

model:
sets:
var/1..5/;

```

```

link(var,var):c,x;
endsets
data:
c=12 7 9 7 9
8 10 6 6 6
7 17 12 14 9
15 14 6 6 10
4 10 7 7 9;
enddata
min=@sum(link:c*x);
@for(var(i):@sum(var(j):x(i,j))=1);
@for(var(j):@sum(var(i):x(i,j))=1);
@for(link:@bin(x));
end

```

**例 2.4** 已知 10 个商业网点的坐标如表 2.2 所示，现要在 10 个网点中选择适当位置设置供应站，要求供应站只能覆盖 10 公里之内的网点，且每个供应站最多供应 5 个网点，如何设置才能使供应站的数目最小，并求最小供应站的个数。

表 2.2 商业网点的  $x$  坐标和  $y$  坐标数据

$x$	9.4888	8.7928	11.5960	11.5643	5.6756	9.8497	9.1756	13.1385	15.4663	15.5464
$y$	5.6817	10.3868	3.9294	4.4325	9.9658	17.6632	6.1517	11.8569	8.8721	15.5868

解 记  $d_{ij}$  ( $i=1,\dots,10$ ) 表示第  $i$  个营业网点与第  $j$  个营业网点之间的距离，引进 0-1 变量

$$x_i = \begin{cases} 1, & \text{第 } i \text{ 个网点建立供应站,} \\ 0, & \text{第 } i \text{ 个网点不建立供应站.} \end{cases}$$

$$y_{ij} = \begin{cases} 1, & \text{第 } j \text{ 个网点被第 } i \text{ 个网点的供应站覆盖,} \\ 0, & \text{否则.} \end{cases}$$

建立如下的 0-1 整数规划模型

$$\begin{aligned}
& \min \sum_{i=1}^n x_i \\
& \text{s.t.} \begin{cases} \sum_{i=1}^n y_{ij} \geq 1, & j=1,2,\dots,n, \\ d_{ij} y_{ij} \leq 10 x_i, & i,j=1,2,\dots,n, \\ x_i \geq y_{ij}, & i,j=1,2,\dots,n, \\ \sum_{j=1}^{10} y_{ij} \leq 5, & i=1,2,\dots,10, \\ x_i, y_{ij} = 0 \text{ 或 } 1, & i,j=1,2,\dots,n. \end{cases}
\end{aligned}$$

计算的LINGO程序如下

```

model:
sets:
num/1..10/:x0,y0,x;
link(num,num):y,d;
endsets
data:
x0=9.4888   8.7928   11.5960   11.5643   5.6756   9.8497   9.1756   13.1385   15.4663
          15.5464;
y0=5.6817   10.3868   3.9294   4.4325   9.9658   17.6632   6.1517   11.8569   8.8721

```

```

15.5868;
enddata
calc:
@for(link(i,j):d(i,j)=@sqrt((x0(i)-x0(j))^2+(y0(i)-y0(j))^2));
endcalc
min=@sum(num:x);
@for(num(j):@sum(num(i):y(i,j))>1);
@for(link(i,j):d(i,j)*y(i,j)<10*x(i));
@for(link(i,j):x(i)>y(i,j));
@for(num(i):@sum(num(j):y(i,j))<5);
@for(num:@bin(x));
@for(link:@bin(y));

```

### 2.1.3 非线性规划

建模时尽量建立线性规划模型，或者把非线性规划模型线性化。

例 2.5 求解下列规划问题

$$\begin{aligned}
\min \quad & z = |x_1| + 2|x_2| + 2|x_3| + 4|x_4|, \\
\text{s.t.} \quad & x_1 - x_2 - x_3 + x_4 = 0, \\
& x_1 - x_2 + x_3 - 3x_4 = 1, \\
& x_1 - x_2 - 2x_3 + 3x_4 = \frac{1}{2}.
\end{aligned}$$

解一：直接用 LINGO 软件求解，LINGO 程序如下

```

model:
sets:
row/1..3/:b;
col/1..4/:c,x;
link(row,col):a;
endsets
data:
c=1 2 2 4;
b=0 1 -0.5;
a=1 -1 -1 1 1 -1 1 -3 1 -1 -2 3;
enddata
min=@sum(col:c*@abs(x));
@for(row(i):@sum(col(j):a(i,j)*x(j))=b(i));
@for(col:@free(x));
end

```

解二 先线性化，做变量变换  $u_i = \frac{x_i + |x_i|}{2} \geq 0$ ,  $v_i = \frac{|x_i| - x_i}{2} \geq 0$ ,  $i = 1, 2, 3, 4$ ，记  $x = [x_1, x_2, x_3, x_4]^T$ ,  $u = [u_1, u_2, u_3, u_4]^T$ ,  $v = [v_1, v_2, v_3, v_4]^T$ ,  $|x| = [|x_1|, |x_2|, |x_3|, |x_4|]^T$ ，则  $x = u - v$ ,  $|x| = u + v$ ，则可把模型变换为线性规划模型

$$\begin{aligned}
\min \quad & c^T(u + v), \\
\text{s.t.} \quad & \begin{cases} A(u - v) = b \\ y \geq 0. \end{cases}
\end{aligned}$$

其中  $c = [1, 2, 2, 4]^T$ ,  $b = [0, 1, -\frac{1}{2}]^T$ ,  $A = \begin{bmatrix} 1 & -1 & -1 & 1 \\ 1 & -1 & 1 & -3 \\ 1 & -1 & -2 & 3 \end{bmatrix}$ 。

求解的 LINGO 程序如下

```

model:
sets:
row/1..3/:b;
col/1..4/:c,u,v,x;
link(row,col):a;
endsets
data:
c=1 2 2 4;
b=0 1 -0.5;
a=1 -1 -1 1 1 -1 1 -3 1 -1 -2 3;
enddata
min=@sum(col:c*(u+v));
@for(row(i):@sum(col(j):a(i,j)*(u(j)-v(j)))=b(i));
end

```

注 2.1 (1) LINGO 软件可以自动对带有绝对值的数学规划问题进行线性化。

(2) LINGO 线性化时，变量的个数至少扩大为原来的 4 倍，约束条件也增加很多；问题规模大时，可能 LINGO 软件就无法求解了；如果能手工进行线性化的话，尽量手工线性化。

## 2.2 数独问题

数独这个奇特的名字来源于日语 Sudoku，是 18 世纪瑞士数学家欧拉发明的，后在美国发展，并在日本得以发扬光大。Sudoku 的规则十分简单，就是在  $9 \times 9$  的九宫格里面填数字，每个方格中填入合适的数字以使得每行、每列以及每个九宫格都要包含从 1~9 的数字且互不相同。数独的玩法逻辑简单，数字排列方式千变万化，不少教育者认为数独是锻炼脑筋的好方法。谜题中会预先填入若干数字，其他方格为空白，玩家得依谜题中的数字分布状况，逻辑推敲出剩下的空格里是什么数字。由于规则简单，在推敲之中完全不必用到数学计算，只须用到逻辑推理能力，所以无论男女老幼，人人都可以玩，而且容易上手、容易入迷。世界各地有很多数独俱乐部，还有国家如法国等专门举行过数独比赛，其风靡程度可见一斑。

目前求数独的方法主要有两种，一种是基于计算机的回溯法或类似的全枚举方法，这种方法对小规模的问题还可以，对  $25 \times 25$ ， $36 \times 36$  及更大规模的问题就难以凑效了，而且这种方法没有体现出智能性。另一种方法就是基于人的思维，寻找求解的特殊技巧，如数独终结者软件分别总结了直观法和候选数法两大类，其中直观法有单元唯一法、单元排除法、区域排除法、唯一余数法、组合排除法、矩形排除法，候选数法有显式唯一法、隐式唯一法、区块删除法、显式数对法、显式三数集法、显式四数集法、隐式数对法、隐式三数集法、隐式四数集法、矩形对角线法、XY 形态匹配法、XYZ 形态匹配法、三链数删减法、WXYZ 形态匹配法。这些方法过分注重具体的技巧，缺乏一般性。可以从数独本身具有的性质出发，建立一些求解规则，根据规则设计算法进行求解，既避免了通常求解所使用的特殊技巧，又避免了计算机求解的完全枚举。将人的推理和计算机的枚举能力结合起来，可以有效地提高求解速度。

目前比较流行的数独包括九宫数独、对角数独、数比数独、 $m \times n$  数独、锯齿数独、Killer 数独、Kakuro 数独。九宫数独是最原始和最常见的，将其规模扩大，则是  $m \times n$  数独；如果增加两条对角线的限制，则是对角数独；如果考虑某些位置所填数之间的大小关系，则是数

比数独；如果将九宫格的形状任意改变成不规则形状，则是锯齿数独；如果考虑数在某些区域的和，则是 Killer 数独；如果只考虑行列区域数字和，则是 Kakuro 数独。

下面给出九宫数独问题的线性规划模型，并用 LINGO 软件进行求解。

例 2.6 求如图 2.1 所示的九宫数独问题。

	2			3			4	
6								3
		4					5	
			8		6			
8				1				6
			7		5			
		7				6		
4								8
	3			4			2	

图 2.1 九宫数独问题

解 对一般的九宫数独问题，我们建立如下通用的 0-1 整数规划模型。

设每个格子用  $(i, j)$  ( $i, j = 1, 2, \dots, 9$ ) 表示该空格所在的行和列。引进 0-1 决策变量

$$x_{ijk} = \begin{cases} 1, & \text{空格}(i, j) \text{处填} k, \\ 0, & \text{空格}(i, j) \text{处不填} k, \end{cases} \quad i, j, k = 1, 2, \dots, 9.$$

约束条件分如下 5 类：

(1) 每个空格恰好填一个数字

$$\sum_{k=1}^9 x_{ijk} = 1, \quad i, j = 1, 2, \dots, 9.$$

(2) 每行每个数字恰好填一次

$$\sum_{j=1}^9 x_{ijk} = 1, \quad i, k = 1, 2, \dots, 9.$$

(3) 每列每个数字恰好填一次

$$\sum_{i=1}^9 x_{ijk} = 1, \quad j, k = 1, 2, \dots, 9.$$

(4) 每个九宫格中每个数字恰好填一次

左上角的  $3 \times 3$  九宫格对应的条件为

$$\sum_{i=1}^3 \sum_{j=1}^3 x_{ijk} = 1, \quad k = 1, 2, \dots, 9.$$

类似地，可以写出其他 8 个九宫格的约束条件；全部 9 个九宫格的约束条件可综合为

$$\sum_{i=1}^3 \sum_{j=1}^3 x_{i+u, j+v, k} = 1, \quad u, v \in \{0, 3, 9\}, k = 1, 2, \dots, 9.$$

(4) 初值条件

如果  $(i, j)$  处填入  $k$ ，则有  $x_{ijk} = 1$ 。例如图 2.1 第 1 行对应的初值条件为

$$x_{122} = 1, x_{153} = 1, x_{184} = 1,$$

第 9 行对应的初值条件为

$$x_{923} = 1, x_{954} = 1, x_{982} = 1.$$

求解数独问题，实际上是不需要目标函数，只需求可行解即可。为了利用 LINGO 软件求解方便，我们构造一个虚拟的目标函数

$$\min z = \sum_{i=1}^9 \sum_{j=1}^9 \sum_{k=1}^9 x_{ijk}.$$

显然  $z$  的取值恒等于 81。

综上所述，我们建立通用数独问题的 0-1 整数规划模型：

$$\begin{aligned} \min z &= \sum_{i=1}^9 \sum_{j=1}^9 \sum_{k=1}^9 x_{ijk}, \\ \text{s.t.} \quad &\begin{cases} \sum_{k=1}^9 x_{ijk} = 1, & i, j = 1, 2, \dots, 9, \\ \sum_{j=1}^9 x_{ijk} = 1, & i, k = 1, 2, \dots, 9, \\ \sum_{i=1}^9 x_{ijk} = 1, & j, k = 1, 2, \dots, 9, \\ \sum_{i=1}^3 \sum_{j=1}^3 x_{i+u, j+v, k} = 1, & u, v \in \{0, 3, 9\}, k = 1, 2, \dots, 9, \\ x_{122} = 1, x_{153} = 1, x_{184} = 1, \\ \dots\dots\dots \\ x_{923} = 1, x_{954} = 1, x_{982} = 1, \\ x_{ijk} = 0 \text{ 或 } 1, & i, j, k = 1, 2, \dots, 9. \end{cases} \end{aligned}$$

利用 LINGO 软件，求得数独问题的解如图 2.2 所示。

9	2	5	6	3	1	8	4	7
6	1	8	5	7	4	2	9	3
3	7	4	9	8	2	5	6	1
7	4	9	8	2	6	1	3	5
8	5	2	4	1	3	9	7	6
1	6	3	7	9	5	4	8	2
2	8	7	3	5	9	6	1	4
4	9	1	2	6	7	3	5	8
5	3	6	1	4	8	7	2	9

图 2.2 数独问题的解

计算的 LINGO 程序如下：

```
model:
sets:
num/1..9/;
link(num,num,num):x;
num2/1..21/; !初值个数为21个;
```



```

num3/1..3/:u,v;
link2(num2,num3):a; !描述一个初值需要3个数据, (i,j)处为k;
endsets
data:
a=1,2,2 !该行表示 (1,2) 处填入2;
1,5,3
1,8,4
2,1,6
2,9,3
3,3,4
3,7,5
4,4,8
4,6,6
5,1,8
5,5,1
5,9,6
6,4,7
6,6,5
7,3,7
7,7,6
8,1,4
8,9,8
9,2,3
9,5,4
9,8,2;
u=0,3,6;
v=0,3,6;
enddata
calc:
@for(num2(n):x(a(n,1),a(n,2),a(n,3))=1); !赋决策变量的初值条件;
endcalc
min=@sum(link:x);
@for(num(i):@for(num(j):@sum(num(k):x(i,j,k))=1));
@for(num(i):@for(num(k):@sum(num(j):x(i,j,k))=1));
@for(num(j):@for(num(k):@sum(num(i):x(i,j,k))=1));
@for(num(k):@for(num3(m):@for(num3(n):@sum(num3(i):@sum(num3(j):x(i+u(m),j+v(n),k))=1)))));
@for(link:@bin(x));
end

```

## 2.3 非线性拟合

设已知  $(x, y)$  的观测值  $(x_i, y_i)$ ,  $i = 1, 2, \dots, n$ , 要拟合一个函数  $y = \hat{f}(x)$ , 一般都使用最小二乘准则拟合, 即所拟合的函数使得

$$\sum_{i=1}^n (y_i - \hat{y}_i)^2$$

达到最小值, 这里  $\hat{y}_i = \hat{f}(x_i)$ , 称为最小二乘拟合。

使用最小二乘法进行非线性拟合时, 实际上是求多元函数的极值问题。Matlab 在很多

情形下只能求得一个局部极小值，LINGO 算法比较先进，可能求得全局最优解。

例 2.7 已知  $x, y$  的观测值见表 2.3。用最小二乘法拟合函数  $y = \frac{a}{e^{bx} + c}$ 。

表 2.3  $x, y$  的观测值

$x$	6	2	6	7	4	2	5	9
$y$	4	9	5	3	8	5	8	2

拟合的 LINGO 程序如下

```

model:
sets:
num/1..8/:x0,y0;
endsets
data:
x0=62    6    7    4    2    5    9;
y0=49    5    3    8    5    8    2;
enddata
min=@sum(num:(y0-a/(@exp(b*x0)+c))^2);
@free(a); @free(b); @free(c);
end

```

## 2.4 目标规划

例 2.8 某市政府拟投入一笔资金和一定数量的劳动力建设两类公益项目 A 和 B，目的是方便市民的生活，提高城市的生活质量。根据预测投入 1 万元资金和 1 百个劳动力·h（即每个劳动力用 1h），分别可以建成 1 个项目 A 和两个项目 B。如果投入 1 个劳动力·h 需要支付 10 元，市政府为了用有限的资金和劳动力，并用最快的时间建成这批项目，服务于社会，服务于人民。市政府依次提出下面的四条要求。

- (1) 至少要建 50 个项目 A；
- (2) 至多建设 60 个项目 B；
- (3) 至少要利用 80 万元资金和 10000 个劳动力·h；
- (4) 总投入资金不超过预算 120 万元。

试为该市政府制定一个满意的项目建设方案。

解：设项目 A，B 的建设个数分别为  $x_1$ ， $x_2$ ，各层目标的权重分别为  $p_i$ （ $i=1,2,3,4$ ），建立如下的目标规划模型

$$\begin{aligned}
 & \min p_1 d_1^- + p_2 d_2^+ + p_3 (d_3^- + d_4^-) + p_4 d_5^+, \\
 \text{s. t. } & x_1 + d_1^- - d_1^+ = 50, \\
 & x_2 + d_2^- - d_2^+ = 60, \\
 & 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_3^- - d_3^+ = 800000, \\
 & 100x_1 + 50x_2 + d_4^- - d_4^+ = 10000, \\
 & 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_5^- - d_5^+ = 1200000, \\
 & x_1, x_2 \geq 0 \text{ 且为整数}, \\
 & d_i^-, d_i^+ \geq 0, \quad i=1,2,3,4,5.
 \end{aligned}$$

求解目标规划问题有多种解法，其中的序贯解法是要求解如下的 4 个线性规划问题：

- (1) 第一个线性规划问题

$$\begin{aligned}
 & \min d_1^-, \\
 \text{s. t. } & x_1 + d_1^- - d_1^+ = 50, \\
 & x_2 + d_2^- - d_2^+ = 60,
 \end{aligned}$$

$$\begin{aligned}
10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_3^- - d_3^+ &= 800000, \\
100x_1 + 50x_2 + d_4^- - d_4^+ &= 10000, \\
10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_5^- - d_5^+ &= 1200000, \\
x_1, x_2 &\geq 0 \text{ 且为整数}, \\
d_i^-, d_i^+ &\geq 0, \quad i=1,2,3,4,5.
\end{aligned}$$

求得的目标函数的最优值  $d_1^+ = 0$ 。

(2) 第二个线性规划问题

$$\begin{aligned}
\min \quad & d_2^+, \\
\text{s. t.} \quad & x_1 + d_1^- - d_1^+ = 50, \\
& x_2 + d_2^- - d_2^+ = 60, \\
& 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_3^- - d_3^+ = 800000, \\
& 100x_1 + 50x_2 + d_4^- - d_4^+ = 10000, \\
& 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_5^- - d_5^+ = 1200000, \\
& x_1, x_2 \geq 0 \text{ 且为整数}, \\
& d_i^-, d_i^+ \geq 0, \quad i=1,2,3,4,5, \\
& d_1^- = 0.
\end{aligned}$$

求得的目标函数的最优值  $d_2^{++} = 0$ 。

(3) 第三个线性规划问题

$$\begin{aligned}
\min \quad & d_3^- + d_4^-, \\
\text{s. t.} \quad & x_1 + d_1^- - d_1^+ = 50, \\
& x_2 + d_2^- - d_2^+ = 60, \\
& 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_3^- - d_3^+ = 800000, \\
& 100x_1 + 50x_2 + d_4^- - d_4^+ = 10000, \\
& 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_5^- - d_5^+ = 1200000, \\
& x_1, x_2 \geq 0 \text{ 且为整数}, \\
& d_i^-, d_i^+ \geq 0, \quad i=1,2,3,4,5, \\
& d_1^- = 0, \quad d_2^+ = 0.
\end{aligned}$$

求得的目标函数的最优值  $(d_3^- + d_4^-)^* = 0$ 。

(4) 第四个线性规划问题

$$\begin{aligned}
\min \quad & d_5^+, \\
\text{s. t.} \quad & x_1 + d_1^- - d_1^+ = 50, \\
& x_2 + d_2^- - d_2^+ = 60, \\
& 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_3^- - d_3^+ = 800000, \\
& 100x_1 + 50x_2 + d_4^- - d_4^+ = 10000, \\
& 10000x_1 + 5000x_2 + 10(100x_1 + 50x_2) + d_5^- - d_5^+ = 1200000, \\
& x_1, x_2 \geq 0 \text{ 且为整数}, \\
& d_i^-, d_i^+ \geq 0, \quad i=1,2,3,4,5, \\
& d_1^- = 0, \quad d_2^+ = 0, \quad d_3^- + d_4^- = 0.
\end{aligned}$$

求得的目标函数的最优值  $d_5^{++} = 0$ ，求得的满意解为  $x_1 = 70$ ， $x_2 = 60$ 。

计算的 LINGO 程序如下。

```

model:
sets:
level/1..4/:p,z,goal;
variable/1..2/:x;
s_con_num/1..5/:g,dplus,dminus;

```

```

s_con(s_con_num,variable):a;
obj(level,s_con_num)/1 1,2 2,3 3,3 4,4 5/:wplus,wminus;
endsets
data:
ctr=?;
goal=? ? ? 0;
g=50 60 800000 10000 1200000;
a=1 0 0 1 11000 5500 100 50 11000 5500;
wplus=0 1 0 0 1;
wminus=1 1 1 1 0;
enddata
min=@sum(level:p*z);
p(ctr)=1;
@for(level(i)|i#ne#ctr:p(i)=0);
@for(level(i):z(i)=@sum(obj(i,j):wplus(i,j)*dplus(j)+wminus(i,j)*dminus(j)));
@for(s_con_num(i):@sum(variable(j):a(i,j)*x(j))+dminus(i)-dplus(i)=g(i));
@for(level(i)|i #lt# @size(level):@bnd(0,z(i),goal(i)));
@for(variable:@gin(x));
end

```

上述LINGO程序需要运行4次:

第一次运行ctr值输入1, goal(1), goal(2), goal(3)都输入为100000 (一个充分大的正实数)。求得的目标函数的最优值0作为下一次运行goal(1)的值。

第二次运行ctr值输入2, goal(1), goal(2), goal(3)分别输入为0, 100000, 100000。求得的目标函数的最优值0作为下一次运行goal(2)的值。

第三次运行ctr值输入3, goal(1), goal(2), goal(3)分别输入为0, 0, 100000。求得的目标函数的最优值0作为下一次运行goal(3)的值。

第四次运行ctr值输入4, goal(1), goal(2), goal(3)分别输入为0。

我们也可以利用 LINGO 的子模型计算, 运行一次即可以得到计算结果, 计算的 LINGO 程序如下:

```

model:
sets:
level/1..4/:p,z,goal;
variable/1..2/:x;
s_con_num/1..5/:g,dplus,dminus;
s_con(s_con_num,variable):a;
obj(level,s_con_num)/1 1,2 2,3 3,3 4,4 5/:wplus,wminus;
endsets
data:
goal=100000;
g=50 60 800000 10000 1200000;
a=1 0 0 1 11000 5500 100 50 11000 5500;
wplus=0 1 0 0 1;
wminus=1 1 1 1 0;
enddata
submodel mubiao:
[mobj]min=@sum(level:p*z);
p(ctr)=1;
@for(level(i)|i#ne#ctr:p(i)=0);
@for(level(i):z(i)=@sum(obj(i,j):wplus(i,j)*dplus(j)+wminus(i,j)*dminus(j)));
@for(s_con_num(i):@sum(variable(j):a(i,j)*x(j))+dminus(i)-dplus(i)=g(i));

```

```

@for(level(i)|i #lt# @size(level):@bnd(0,z(i),goal(i)));
@for(variable:@gin(x));
endsubmodel
calc:
@for(level(i): ctr=i; @solve(mubiao); goal(i)= mobj; @write(' 第',ctr,' 次 运 算 :
x(1)=' , x(1) , ' , x(2)=' , x(2) , ' , 最优偏差值为' , mobj , @newline(2) ) );
endcalc
end

```

## 2.5 数据包络分析

下面我们LINGO子函数的编程来实现数据包络分析的计算。

1978年A. Charnes, W. W. Cooper和E. Rhodes给出了评价多个决策单元 (Decision Making Units, 简称DMU) 相对有效性的数据包络分析方法 (data envelopment analysis, DEA)。

目前, 数据包络分析是评价具有多指标输入和多指标输出系统的较为有效的方法。

数据包络分析有多种模型, 其中C<sup>2</sup>R (由Charnes, Cooper和Rhodes三位作者的第一个英文字母命名) 的建模思路清晰、模型形式简单、理论完善。设有  $n$  个DMU, 每个DMU都有  $m$  种投入和  $s$  种产出, 设  $x_{ij}$  ( $i=1, \dots, m, j=1, \dots, n$ ) 表示第  $j$  个DMU的第  $i$  种投入量,  $y_{rj}$  ( $r=1, \dots, s, j=1, \dots, n$ ) 表示第  $j$  个DMU的第  $r$  种产出量,  $v_i$  ( $i=1, \dots, m$ ) 表示第  $i$  种投入的权值,  $u_r$  ( $r=1, \dots, s$ ) 表示第  $r$  种产出的权值。

向量  $X_j, Y_j (j=1, \dots, n)$  分别表示决策单元  $j$  的输入和输出向量,  $v$  和  $u$  分别表示输入、输出权值向量, 则  $X_j = (x_{1j}, x_{2j}, \dots, x_{mj})^T$ ,  $Y_j = (y_{1j}, y_{2j}, \dots, y_{sj})^T$ ,  $u = (u_1, u_2, \dots, u_m)^T$ ,  $v = (v_1, v_2, \dots, v_s)^T$ 。

定义决策单元  $j$  的效率评价指数为

$$h_j = (u^T Y_j) / (v^T X_j), \quad (j=1, \dots, n).$$

评价决策单元  $j_0$  效率的数学模型为

$$\begin{aligned} \max \quad & \frac{u^T Y_{j_0}}{v^T X_{j_0}}, \\ \text{s.t.} \quad & \begin{cases} \frac{u^T Y_j}{v^T X_j} \leq 1, & j=1, 2, \dots, n, \\ u \geq 0, v \geq 0, u \neq 0, v \neq 0. \end{cases} \end{aligned} \quad (1)$$

通过Charnes-Cooper变换:  $\omega = tv$ ,  $\mu = tu$ ,  $t = \frac{1}{v^T X_{j_0}}$ , 可以将模型(1)变化为等价的线性规划问题

$$\begin{aligned} \max \quad & V_{j_0} = \mu^T Y_{j_0}, \\ \text{s.t.} \quad & \begin{cases} \omega^T X_j - \mu^T Y_j \geq 0, & j=1, 2, \dots, n, \\ \omega^T X_{j_0} = 1, \\ \omega \geq 0, \mu \geq 0. \end{cases} \end{aligned} \quad (2)$$

**例 2.9 (多指标评价问题)** 某市教委需要对六所重点中学进行评价, 其相应的指标如表 2.4 所示。表 2.4 中的生均投入和非低收入家庭百分比是输入指标, 生均写作得分和生均科技得分是输出指标。请根据这些指标, 评价哪些学校是相对有效的。

表 2.4 评价指标数据表

学 校	A	B	C	D	E	F
-----	---	---	---	---	---	---

生均投入(百元/年)	89.39	86.25	108.13	106.38	62.40	47.19
非低收入家庭百分比(%)	64.3	99	99.6	96	96.2	79.9
生均写作得分(分)	25.2	28.2	29.4	26.4	27.2	25.2
生均科技得分(分)	223	287	317	291	295	222

```

model:
sets:
    dmu/1..6/:s,t,p;      !决策单元（或评价对象），s,t为中间变量;
    inw/1..2/:omega;      !输入权重;
    outw/1..2/:mu;        !输出权重;
    inv(inw,dmu):x;        !输入变量;
    outv(outw,dmu):y;
endsets
data:
    x=89.39      86.25      108.13      106.38      62.40      47.19
        64.3      99        99.6      96          96.2      79.9;
    y=25.2      28.2      29.4      26.4      27.2      25.2
        223      287      317      291      295      222;
enddata
submodel subopt:
    max=@sum(dmu:p*t);
    p(flag)=1;
    @for(dmu(i)|i#ne#flag:p(i)=0);
    @for(dmu(j):s(j)=@sum(inw(i):omega(i)*x(i,j)));
    t(j)=@sum(outw(i):mu(i)*y(i,j));s(j)>t(j));
    @sum(dmu:p*s)=1;
endsubmodel
calc:
    @for(dmu(k): flag=k;
    @solve(subopt));
endcalc
end

```

## 习题 2

### 2.1 求解下列非线性整数规划问题

$$\max z = x_1^2 + x_2^2 + 3x_3^2 + 4x_4^2 + 2x_5^2 - 8x_1 - 2x_2 - 3x_3 - x_4 - 2x_5,$$

$$\text{s.t.} \begin{cases} 0 \leq x_i \leq 99, \text{且} x_i \text{为整数} (i=1, \dots, 5), \\ x_1 + x_2 + x_3 + x_4 + x_5 \leq 400, \\ x_1 + 2x_2 + 2x_3 + x_4 + 6x_5 \leq 800, \\ 2x_1 + x_2 + 6x_3 \leq 200, \\ x_3 + x_4 + 5x_5 \leq 200. \end{cases}$$

### 2.2 求解下列非线性规划:

$$\max z = \sum_{i=1}^{100} \sqrt{x_i},$$

$$\text{s.t.} \begin{cases} x_1 \leq 10, \\ x_1 + 2x_2 \leq 20, \\ x_1 + 2x_2 + 3x_3 \leq 30, \\ x_1 + 2x_2 + 3x_3 + 4x_4 \leq 40, \\ \sum_{i=1}^{100} (101-i)x_i \leq 1000, \\ x_i \geq 0, i = 1, 2, \dots, 100. \end{cases}$$

2.3 某股民决定对 6 家公司的股票进行投资，根据对这 6 家公司的了解，估计了这 6 家公司股票的明年预期收益和这 6 种股票收益的协方差矩阵的数据见表 2.5。要获得至少 25% 的预期收益，最小风险是多少？

表 2.5 公司股票明年预期收益和收益的协方差矩阵数据

股票	收益率/%	协方差					
		公司 1	公司 2	公司 3	公司 4	公司 5	公司 6
公司 1	20	0.032	0.005	0.03	-0.031	-0.027	0.01
公司 2	42	0.005	0.1	0.085	-0.07	-0.05	0.02
公司 3	100	0.03	0.085	0.333	-0.11	-0.02	0.042
公司 4	50	-0.031	-0.07	-0.11	0.125	0.05	-0.06
公司 5	46	-0.027	-0.05	-0.02	0.05	0.065	-0.02
公司 6	30	0.01	0.02	0.042	-0.06	-0.02	0.08

2.4 有 4 名同学到一家公司参加三个阶段的面试：公司要求每个同学都必须首先找公司秘书初试，然后到部门主管处复试，最后到经理处参加面试，并且不允许插队（即在任何一个阶段 4 名同学的顺序是一样的）。由于 4 名同学的专业背景不同，所以每人在三个阶段的面试时间也不同，如表 2.6 所示。这 4 名同学约定他们全部面试完以后一起离开公司。假定现在时间是早晨 8:00，请问他们最早何时能离开公司？

表 2.6 面试时间要求

	秘书初试	主管复试	经理面试
同学甲	14	16	21
同学乙	19	17	10
同学丙	10	15	12
同学丁	9	12	13

2.5 求图 2.3 所示的九宫数独问题。

1					7		9	
	3			2				8
		9	6			5		
		5	3			9		
	1			8				2
6					4			
3							1	
	4							7
		7				3		

图 2.3 九宫数独问题

2.6 利用表2.7的数据，拟合函数  $y = a \sin(x_1) + e^{bx_2} + \cos(cx_3)$ 。

表 2.7  $x_1, x_2, x_3, y$  的观测值

$x_1$	6	2	6	7	4	2	5	9
$x_2$	4	9	5	3	8	5	8	2
$x_3$	2	5	6	3	6	6	8	7
$y$	5	2	1	9	7	4	3	3

2.7 试求解多目标线性规划问题。

$$\max z_1 = 100x_1 + 90x_2 + 80x_3 + 70x_4,$$

$$\min z_2 = 3x_2 + 2x_4,$$

$$\text{s.t.} \begin{cases} x_1 + x_2 \geq 30, \\ x_3 + x_4 \geq 30, \\ 3x_1 + 2x_3 \leq 120, \\ 3x_2 + 2x_4 \leq 48, \\ x_i \geq 0, \quad i = 1, 2, 3, 4. \end{cases}$$